

УДК 532.4

АППРОКСИМАЦИЯ ФУНКЦИИ ДЕБАЯ $D_3(x)$ В ОБЛАСТИ $0 \leq x < \infty$ ФУНКЦИЕЙ КЛАССА C^2

Д. Г. Гордеев, А. И. Ломайкин
(РФЯЦ-ВНИИЭФ)

Представлены результаты подбора и реализации аппроксимационного выражения для функции Дебая $D_3(x)$ в области $0 \leq x < \infty$. Аппроксимация обеспечивает непрерывность включительно до второй производной и приемлемое для прикладных программ время расчета.

Введение

Для описания термодинамических свойств кристаллов и твердотельных фаз различных веществ в рамках равновесной термодинамики используется приближение Дебая (см., например, [1]). В классической теории Дебая часть свободной энергии вещества, связанная с тепловыми колебаниями его атомной структуры, описывается соотношением $F(\delta, T) = C_V T D_3\left(\frac{\theta(\delta)}{T}\right)$, где C_V — теплоемкость вещества; δ — удельное сжатие; T — термодинамическая температура; $\theta(\delta)$ — температура Дебая; $D_3\left(\frac{\theta(\delta)}{T}\right)$ — функция Дебая (в дальнейшем используется обозначение $D(x)$):

$$D(x) = \frac{3}{x^3} \int_0^x \frac{t^3 dt}{e^t - 1}, \quad x \geq 0. \quad (1)$$

Подход, основанный на приближении Дебая, используется для описания термодинамических свойств в ряде моделей уравнений состояния (см., например, [2, 3]).

Для получения значений этой функции и ее производных необходимо вычисление интеграла с переменным верхним пределом. При многократном вычислении $D(x)$ использование представления (1) для нее нецелесообразно, так как сильно увеличивает время счета. Оптимальным выходом является аппроксимация с заданной точностью функции Дебая и ее производных. Функцию не удастся представить единым аналитическим выражением, позволяющим проводить вычисления во всем диапазоне изменения аргумента. Известны удобные для вычислений аналитические представления функции Дебая, полученные для больших и малых значений аргумента (см., например, [4]).

В данной работе предлагается аппроксимация функции Дебая, не содержащая разрывов в функции и ее первой и второй производных во всей области изменения аргумента. Для построения аппроксимирующей функции используется метод, описанный в работах [5, 6]; отличия заключаются лишь в том, что на каждом из сегментов, на которые поделена числовая ось, используются другие аппроксимирующие функции. В приложении к статье приводится текст программы, реализующей предложенную аппроксимацию. Алгоритм расчета и код программы оптимизированы с целью уменьшения затрат машинного времени. Следует отметить, что программа позволяет рассчитывать значения второй производной, необходимые для вычисления термодинамических функций в ряде уравнений состояния (см., например, [7]).

Свойства функции Дебая и ее асимптотики

Первая и вторая производные функции Дебая связаны с самой функцией следующими соотношениями:

$$D'(x) = \frac{3}{e^x - 1} - \frac{3}{x}D(x); \quad (2)$$

$$D''(x) = \frac{12}{x^2}D(x) - \frac{3e^x}{(e^x - 1)^2} - \frac{9}{x} \frac{1}{e^x - 1}. \quad (3)$$

Вторую производную также можно выразить через функцию и первую производную:

$$D''(x) = \frac{1}{x^2} \left[12D(x) - G(x) \left(3 + x + \frac{G(x)}{3} \right) \right], \quad (4)$$

где $G(x) = xD'(x) + 3D(x)$.

Область определения функции $0 \leq x < \infty$ разбивается на пять участков, как показано на рис. 1. На сегменте 2 проводится сплайн-аппроксимация функции Дебая. На других участках используются аналитические асимптотики, простые в вычислительном отношении, справедливые при относительно малых (сегмент 1) и больших ($x > 5$) значениях аргумента. Участки 3–5 введены для экономии затрат машинного времени при вычислении за счет использования различного числа членов ряда, являющегося асимптотикой функции (1) при больших значениях аргумента. Разбиение области изменения аргумента на участки проводится и в работах [5, 6]. Отличие подхода, описанного в данной работе, состоит в использовании большего количества участков и применении другого типа сплайна.




Рис. 1. Разбиение области определения функции Дебая

Функцию (1) можно представить в виде ряда, сходящегося при $x < 2\pi$ [4]:

$$D(x) = 1 - \frac{3}{8}x + 3 \sum_{k=1}^{\infty} \frac{B_{2k}}{(2k+3)(2k)!} \cdot x^{2k},$$

где B_{2k} — числа Бернулли. Для описания функции на отрезке 1 (см. рис. 1) используются три первых члена данного ряда:

$$F_1(x) = 1 - \frac{3}{8}x + \frac{x^2}{20} - \frac{1}{7240}x^4. \quad (5)$$

Для остаточного члена справедлива оценка $R(x) = |D(x) - F_1(x)| < \frac{x^6}{90720}$. На отрезке 1 относительная погрешность аппроксимации функции $\frac{R(x)}{D(x)} \cdot 100\% \leq \frac{R(0,2)}{D(0,2)} \cdot 100\% \approx 7,6 \cdot 10^{-8}\%$. Первая производная функции Дебая вычисляется дифференцированием выражения (5), вторая производная при $x > 10^{-3}$ — по выражению (4), а при $x \leq 10^{-3}$ используется вторая производная от выражения (5). На отрезке 1 абсолютное значение относительной погрешности аппроксимации первой производной не превышает $6 \cdot 10^{-6}\%$.

На отрезках 3–5 функция (1) аппроксимируется другим рядом, сходящимся при $x > 0$ [4]:

$$F_2(x) = \frac{\pi^4}{5x^3} - 3 \sum_{k=1}^n \frac{1}{k} \left(1 + \frac{3}{xk} + \frac{6}{(xk)^2} + \frac{6}{(xk)^3} \right) e^{-kx}. \quad (6)$$

Выражение (6) используется на отрезке 3 при $n = 2$, на отрезке 4 при $n = 1$, на полупрямой 5 при $n = 0$. Первая и вторая производные функции Дебая на отрезках 3 и 4 вычисляются по формулам (2) и (3) соответственно, на полупрямой 5 производные получаются дифференцированием (6).

Для остаточного члена на отрезках 3–5 при $n \geq 1$ справедлива завышенная оценка

$$R_n(x) = |D(x) - F_2(x)| = 3 \sum_{k=n+1}^{\infty} \frac{1}{k} \left(1 + \frac{3}{xk} + \frac{6}{(xk)^2} + \frac{6}{(xk)^3} \right) e^{-kx} < 3 \sum_{k=n+1}^{\infty} e^{-kx} = 3 \frac{e^{-nx}}{e^x - 1}.$$

На отрезке 3 абсолютное значение относительной погрешности аппроксимации функции $\frac{R_2(x)}{D(x)} \cdot 100\% \leq \frac{R_2(5)}{D(5)} \cdot 100\% \approx 7,9 \cdot 10^{-4}\%$. Численная оценка максимального значения этой погрешности $3,2 \cdot 10^{-4}\%$.

На отрезке 4 абсолютное значение относительной погрешности аппроксимации функции $\frac{R_1(x)}{D(x)} \cdot 100\% \leq \frac{R_1(14)}{D(14)} \cdot 100\% \approx 3 \cdot 10^{-8}\%$. Численная оценка максимального значения этой погрешности $1,6 \cdot 10^{-8}\%$.

На полупрямой 5 абсолютное значение относительной погрешности аппроксимации функции $\frac{R_0(x)}{D(x)} \cdot 100\% \leq \frac{R_0(34)}{D(34)} \cdot 100\% \approx 10^{-9}\%$. Численная оценка максимального значения этой погрешности $\approx 10^{-9}\%$.

Для аппроксимации функции $D(x)$ на отрезке 2 используется интерполяционный кубический сплайн, принадлежащий классу дважды непрерывно-дифференцируемых функций (см., например, [8]). Отрезок 2 равномерно разбит на $N = 16$ участков $[x_i, x_{i+1}]$, на каждом из которых сплайн имеет вид

$$F_3(x) = f_i + (x - x_i)[m_i + t(B + tA)], \quad 0 \leq i \leq N, \quad (7)$$

где $A = m_{i+1} - m_i - 2 \frac{f_{i+1} - f_i}{x_{i+1} - x_i}$; $B = \frac{f_{i+1} - f_i}{x_{i+1} - x_i} - A - m_i$; $t = \frac{x - x_i}{x_{i+1} - x_i}$. В этих выражениях f_i ($1 \leq i \leq N - 1$) — точные значения функции Дебая во внутренних узлах сплайна; m_i — приближенные значения первой производной в тех же узлах (получены решением трехдиагональной системы линейных уравнений методом прогонки). В качестве краевых условий (при определении значений m_i) используются значения первых производных на концах отрезка 2: $m_0 = F'_1(x_0)$; $m_N = F'_2(x_N)$. Для того чтобы при аппроксимации сохранялась непрерывность функции Дебая и ее первой производной на концах отрезка 2, используются условия сшивки $f_0 = F_1(x_0)$; $f_N = F_2(x_N)$. Внутри отрезка первая производная функции вычисляется дифференцированием выражения (7), вторая производная — по выражению (4). Использование сплайна в виде (7) обусловлено тем, что это выражение содержит наименьшее количество арифметических операций [8]. Равномерное расположение узлов позволяет быстро определить нужный для вычислений по (7) участок $[x_i, x_{i+1}]$ при заданном значении x . В программе для вычисления $D(x)$ достаточно хранить только значения функции и ее производных m_i в узловых точках.

Для оценки остаточного члена при приближении функции кубическим сплайном имеются выражения [8]

$$R(x) = |D(x) - F_3(x)| = \frac{5}{384} h^4 \max |D^{(4)}(x)|; \quad R'(x) = |D'(x) - F'_3(x)| = \frac{1}{24} h^3 \max |D^{(4)}(x)|,$$

где x принадлежит отрезку интерполяции; $h = x_{i+1} - x_i$. На отрезке 2 абсолютные значения относительных погрешностей сплайн-аппроксимации функции и ее производной составляют $\frac{R(x)}{D(x)} \cdot 100\% < < 10^{-3}\%$ и $\frac{R'(x)}{D'(x)} \cdot 100\% < 10^{-2}\%$. Численная оценка максимальных значений этих погрешностей $3,2 \cdot 10^{-4}$ и $4 \cdot 10^{-3}$ соответственно.

Сплайн-аппроксимацию можно было бы использовать на всем отрезке $[0, 5]$ (не применять выражение (5) на участке 1). Однако такой подход при одной и той же точности описания функции и ее производных требует большего числа узлов сплайна.

Таким образом, для каждого участка, показанного на рис. 1, выбран свой способ аппроксимации функции Дебая (1) и ее первой производной. Во всей области определения функции Дебая при аппроксимации сохраняется непрерывность самой функции и ее первой и второй производных.

Точность аппроксимации

Оценка относительной погрешности $\delta(D(x)) = \frac{D(x) - D(x)_{\text{эталон}}}{D(x)_{\text{эталон}}} \cdot 100\%$ проводилась путем сравнения результатов аппроксимации функции Дебая и ее первой и второй производных с *точными* значениями, полученными с помощью вычисления интеграла в выражении (1) с точностью не менее 10 значащих цифр (значения первой и второй производных вычислялись по формулам (2) и (3) соответственно).

На рис. 2 представлены относительные отклонения (в процентах) значений функции Дебая, вычисленных с использованием аппроксимации, описанной в данной работе, и аппроксимаций [6], от значений, полученных с помощью интегрирования. Максимальное отклонение для функции из работы [6] наблюдается в точке $x \approx 4,45$ и составляет $\approx 10^{-3}\%$. Максимальное отклонение аппроксимации функции Дебая, описанной в данной работе, наблюдается в точке $x \approx 5$ и составляет $\approx -3 \cdot 10^{-4}\%$.

Таким образом, аппроксимация, предложенная авторами, обеспечивает описание функции Дебая с точностью не менее 5 значащих цифр.

На рис. 3 представлены относительные отклонения (в процентах) значений первой производной функции Дебая, вычисленных с использованием аппроксимации, описанной в данной работе. Максимальное отклонение аппроксимации производной наблюдается в точке $x \approx 4,9$ и составляет $\approx 4 \cdot 10^{-3}\%$. Таким образом, предложенная аппроксимация обеспечивает описание первой производной функции Дебая с точностью не менее 4 значащих цифр.

На рис. 4 представлены относительные отклонения (в процентах) значений второй производной функции Дебая, вычисленных с использованием аппроксимации, описанной в данной работе. Мак-




Рис. 2. Относительная погрешность аппроксимации функции Дебая: — — из данной работы; - - - - из [6]




Рис. 3. Относительная погрешность аппроксимации, предложенной в данной работе, первой производной функции Дебая




Рис. 4. Относительная погрешность аппроксимации, предложенной в данной работе, второй производной функции Дебая

симальное отклонение аппроксимации второй производной составляет $\approx 7 \cdot 10^{-2} \%$ при $x = 0,2$. Предложенная аппроксимация обеспечивает описание второй производной функции Дебая с точностью не менее 3 значащих цифр.

Построенная аппроксимация позволяет без разрывов описать функцию и первую производную. Непрерывность описания второй производной достигается за счет вычисления ее по формуле (4).

Результаты оценки времени вычисления

Тестирование на время вычисления проводилось для варианта программы с вычислением только функции Дебая. Для тестирования была выбрана сетка, содержащая 5 участков: $[0; 0,2)$, $[0,86; 4,44)$, $[7,1; 14)$, $[14,1; 34)$, $[34,1; 35,1)$. Каждый участок равномерно разбивался на 10^4 точек. Расчет на заданной сетке проводился 10^4 раз, общее число вызовов составило 10^8 .

Для исключения влияния на время вычисления служебных операций (циклического повторения, вычисления значений аргумента функции на сетке) тестирование организовывалось следующим образом. Сначала определялось время вычисления всех исследуемых аппроксимаций на заданной сетке при заданном количестве повторений с учетом всех служебных операций (базовое время t_6). Затем из вычислений исключалась одна из исследуемых функций и определялось новое время, также с учетом всех служебных операций (время с исключением t_n). Тогда время, затраченное на вычисление исключенной функции, есть разность базового времени и времени с исключением: $t_{\text{расч}} = t_6 - t_n$.

Проведенные исследования показали, что время вычисления по предложенной аппроксимации приблизительно в 1,47 раза меньше времени вычисления по аппроксимации из работы [6], авторы которой не ставили своей целью построение оптимизированного вычислительного модуля.

При использовании выражений (2), (3) и (6) при $n = 1, 2$ требуются вычисления экспоненты, что значительно увеличивает затраты машинного времени. Поэтому при значениях аргумента $x < 34$ используется аппроксимация Паде для экспоненты [9], позволяющая в указанном диапазоне сократить затраты времени ЭВМ, не снижая точности описания исходной функции. Выражение

$$e^x \approx \frac{12(x^2 + 10) + x(x^2 + 60)}{12(x^2 + 10) - x(x^2 + 60)} \quad (8)$$

на интервале $|x| < 0,5 \ln 2$ обеспечивает точность описания экспоненциальной функции не ниже 10^{-8} .

Используя замену переменных $y = x/2^l$, $e^x = \prod_{k=1}^{2^l} e^y$, можно расширить интервал применимости формулы (8) до необходимых пределов ($x < 34$, см. рис. 1). В данной работе использовано значение $l = 5$.

Максимальная погрешность вычисления экспоненциальной функции находится на границах аппроксимируемого интервала и составляет $|\delta(e^x)| \approx 6 \cdot 10^{-2} \%$, т. е. данная аппроксимация обеспечивает описание экспоненциальной функции с совпадением не менее 3 значащих цифр. Использование аппроксимации Паде для экспоненты [9] позволило сократить время вычисления по программе, реализующей предложенную аппроксимацию, примерно на 12 % по сравнению с использованием стандартной функции экспоненты, встроенной в Fortran 90.

Приложение. Программа, реализующая предложенную аппроксимацию функции Дебая

```

subroutine Debay3(t,Dbx,Dx,Dxx)
implicit real*8 (a-h,o-z)
parameter (Pi4=19.481818206800487d0,m=16)
parameter (a0=2.0d-01,b0=0.5d+01)
dimension y(0:m),s(0:m)
data y / 0.9269990476190476D+00, 0.8249629689762340D+00,
*      0.7317590427532673D+00, 0.6471475685537391D+00,
*      0.5707931370319482D+00, 0.5022797638202503D+00,
*      0.4411284737296280D+00, 0.3868159061407920D+00,

```

```

*      0.3387926100035671D+00, 0.2964999308058749D+00,
*      0.2593847027844887D+00, 0.2269112918634890D+00,
*      0.1985708313555613D+00, 0.1738877434731498D+00,
*      0.1524238050351301D+00, 0.1337801211990721D+00,
*      0.1175977894232090D+00/
data s/ -0.3550190476190476D+00, -0.3252957502132044D+00,
*      -0.2961980001859381D+00, -0.2680662532679921D+00,
*      -0.2411960439552844D+00, -0.2158276182457588D+00,
*      -0.1921401160848829D+00, -0.1702504942092926D+00,
*      -0.1502165443385554D+00, -0.1320430817856560D+00,
*      -0.1156902007096048D+00, -0.1010825047997845D+00,
*      -0.8811849438053157D-01, -0.7667900158148193D-01,
*      -0.6663576249785216D-01, -0.5785417116788590D-01,
*      -0.5020770894981588D-01/
if (t.lt.a0) then
  t2=t*t
  Dbx=1.0d0-t*(0.375d0-t*(0.05d0-5.9523809523809526d-04*t2))
  Dx = -0.375D0+t*(0.1D0-2.380952380952381D-3*t2)
  ax2=1d0/t2
  abe = t*Dx+3*Dbx
  if(t.lt.1d-3) then
    Dxx = 0.1d0-t2*7.142857142857143d-3
  else
    Dxx = ax2*(12d0*Dbx-abe*(3d0+t+abe*0.3333333333333333d0))
  end if
  return
else if (t.lt.b0) then
  h=0.3d0; ah=3.333333333333333d0
  by=(t-a0)*ah
  i=int(by)
  i=min(i,m-1)
  x0=a0+h*i; x1=x0+h
  y0=y(i); y1=y(i+1)
  s0=s(i); s1=s(i+1)
  z=(t-x0)*ah
  um0=ah*(y1-y0)
  A=(s0+s1)-(um0+um0)
  B=um0-A-s0
  um=z*A
  Dbx=y0+(t-x0)*(s0+z*(B+um))
  Dx=s0+z*(B+B+um+um+um)
  ax=1.0d0/t
  ax2=ax*ax
  abe = t*Dx+3*Dbx
  Dxx = ax2*(12d0*Dbx-abe*(3d0+t+abe*0.3333333333333333d0))
  return
else
  ax=1.0d0/t
  ax2=ax*ax
  Dbx=Pi4*ax*ax2
  if (t.lt.14d0) then
    p1=1.0d0+ax*(3.0d0+6.0d0*ax*(1.0d0+ax))
    p2=0.5d0+ax*(0.75d0+ax*(0.75d0+0.375d0*ax))

```

```

be=expmyD(t)
dex=1.0d0/be
Dbx=Dbx-3.0d0*dex*(p1+p2*dex)
abe=1.0d0/(be-1.0d0)
Dx=3.0d0*(abe-Dbx*ax)
Dxx=12d0*Dbx*ax2-3.0d0*abe*(ax+ax+ax+be*abe)
else if (t.lt.34) then
p1=1.0d0+ax*(3.0d0+6.0d0*ax*(1.0d0+ax))
be=expmyD(t)
dex=1.0d0/be
abe=1.0d0/(be-1.0d0)
Dbx=Dbx-3.0d0*dex*p1
Dx=3.0d0*(abe-Dbx*ax)
Dxx=12d0*Dbx*ax2-3.0d0*abe*(ax+ax+ax+be*abe)
else
Dx=-(ax+ax+ax)*Dbx
Dxx=12.0d0*ax2*Dbx
end if
end if
contains
real*8 function expmyD(a)
implicit real*8 (a-h,o-z)
real*8, parameter :: a1 = 2d0**(-5d0)
real*8, parameter :: k3 = 12d0/a1; k2 = 5d0*k3/a1; k1 = 2d0*k2/a1
e1=(k1+a*(a*(a+k3)+k2))
e2=(k1+a*(a*(k3-a)-k2))
expmyD=e1/e2
expmyD = expmyD*expmyD; expmyD = expmyD*expmyD;
expmyD = expmyD*expmyD; expmyD = expmyD*expmyD;
expmyD = expmyD*expmyD;
return
end function
end

```

Список литературы

1. Базаров И. П., Геворкян Э. В., Николаев П. Н. Термодинамика и статистическая физика. Теория равновесных систем. М.: Издательство МГУ, 1986.
2. Гордеев Д. Г., Гударенко Л. Ф., Куделькин В. Г. Модель уравнения состояния с переменной теплоемкостью ядер и электронов с учетом температуры Дебая. Уравнения состояния бериллия и алюминия // Вопросы атомной науки и техники. Сер. Теоретическая и прикладная физика. 2005. Вып. 1–2. С. 51–64.
3. Гордеев Д. Г., Гударенко Л. Ф., Жерноклетов М. В. и др. Полуэмпирическая модель уравнения состояния металлов РОСА-М. Уравнение состояния алюминия // Там же. Вып. 3. С. 24–36.
4. Справочник по специальным функциям / Под ред. М. Абрамовица, И. Стиган. М.: Наука, 1979.
5. Елисеев Г. М. L-сплайны, порождаемые уравнениями с постоянными коэффициентами // Физическая механика неоднородных сред. Новосибирск: ИТПМ СО АН СССР, 1984. С. 146–153.
6. Елисеев В. Г., Елисеев Г. М. Аппроксимация интегралов с переменными верхними пределами. Вычисление функции Дебая третьего порядка // Вопросы атомной науки и техники. Сер. Математическое моделирование физических процессов. 2006. Вып. 3. С. 59–65.

7. Гударенко Л. Ф., Жерноклетов М. В., Киришинов С. И. и др. Экспериментальные исследования свойств ударно-сжатого карбогала. Уравнения состояния карбогала и оргстекла // ФГВ. 2004. Т. 40, № 3. С. 104–116.
8. Завьялов Ю. С., Квасов Б. И., Мирошниченко В. К. Методы сплайн-функций. М.: Наука, 1980.
9. Амосов А. А., Дубинский Ю. А., Копченова Н. В. Вычислительные методы для инженеров. М.: Высшая школа, 1994.

Статья поступила в редакцию 10.10.07.
